Maple-assisted proof of formula for A204711

Robert Israel

11 June 2018

The common permanents of the 2×2 blocks could be either 1 or 2. However, the only way the permanents can all be 2 is if all entries are 1. Thus $a(n) = b(n) + 1$, where $b(n)$ counts the cases where the permanents are all 1.
Note that because the sum of coefficients on the right side of the conjectured recurrence
$a(n) = 4\,a(n-1) + 45\,a(n-2) - 126\,a(n-3) - 642\,a(n-4) + 1332\,a(n-5) + 3620\,a(n-6) - 5624\,a(n-7) - 6800\,a(n-8) + 8192\,a(n-9)$
is 1, $a(n)$ satisfies it if and only if $b(n)$ satisfies the same recurrence.

There are $2^7 = 128$ configurations for a row. Consider the 128×128 transition matrix T with entries $T_{ij} = 1$ if the bottom row of a 2×7 sub-array could be in configuration i while the top row is in configuration j (i.e. all 2×2 blocks have permanent 1), and 0 otherwise. The following Maple code computes it. Configurations are encoded as 7-element lists, corresponding to binary digits.

```
> Configs:=  [seq(convert(2^7+i,base,2)[1..7],i=0..2^7-1)]:
> Compatible:= proc(i,j)
   if andmap(k -> Configs[i][k]*Configs[j][k+1]+Configs[i][k+1]*
  Configs[j][k] = 1, [$1..6]) then 1 else 0 fi
  end proc:
> T:= Matrix(128,128,Compatible):
```

Thus $b(n) = u\,T^n v$ where u and v are row and column vectors respectively of all 1's. The following Maple code produces these vectors.

```
> u:= Vector[row](128, 1):
  v:= Vector(128, 1):
```

To check, here are the first few entries of our sequence.

```
> TV[0]:= v:
  for n from 1 to 15 do TV[n]:= T . TV[n-1] od:
> A:= [seq(1+ u . TV[n],n=1..15)];
```

$A := [193, 1361, 8705, 58449, 382849, 2542369, 16748161, 110871041, 731709057, 4838473473, 31954317953, 211206670209, 1395251843713, 9220409667201, 60918293373569]$ **(1)**

Now here is the minimal polynomial P of T, as computed by Maple.

```
> P:= unapply(LinearAlgebra:-MinimalPolynomial(T, t), t);
```

$P := t \mapsto t^{22} - 83\,t^{20} - 64\,t^{19} + 2677\,t^{18} + 3136\,t^{17} - 45779\,t^{16} - 64704\,t^{15} + 465422\,t^{14} + 731840\,t^{13} - 2948686\,t^{12} - 4958976\,t^{11} + 11783248\,t^{10} + 20720128\,t^{9} - 29294608\,t^{8} - 53101056\,t^{7} + 43329264\,t^{6} + 80205568\,t^{5} - 34413888\,t^{4} - 64638464\,t^{3} + 11122432\,t^{2} + 21102592\,t$ **(2)**

This turns out to have degree 22, but with the lowest coefficient 0. Thus we will have

$0 = u\,P(T)\,T^n\,v = \sum_{i=1}^{22} p_i\,b(i+n)$ where p_i is the coefficient of t^i in $P(t)$. That corresponds to a homogeneous linear recurrence of order 21, which would hold true for any u and v, after a delay of 1.

It seems that with our particular u and v we have a recurrence of order only 9, corresponding to a factor of P.

```
> Q:= unapply(t^9 - (4*t^8+45*t^7-126*t^6-642*t^5+1332*t^4+3620*t^3
  -5624*t^2-6800*t+8192),t);
```

$$Q := t \mapsto t^9 - 4\,t^8 - 45\,t^7 + 126\,t^6 + 642\,t^5 - 1332\,t^4 - 3620\,t^3 + 5624\,t^2 + 6800\,t - 8192 \quad \textbf{(3)}$$

The complementary factor $R(t) = \dfrac{P(t)}{Q(t)}$ has degree 13.

```
> R:= unapply(normal(P(t)/Q(t)),t);
```

$$R := t \mapsto \left(t^{12} + 4\,t^{11} - 22\,t^{10} - 98\,t^9 + 149\,t^8 + 858\,t^7 - 222\,t^6 - 3288\,t^5 - 958\,t^4 + 5612\,t^3 + 3220\,t^2 - 3496\,t - 2576\right)\,t \quad \textbf{(4)}$$

Now we want to show that $c(n) = u\,Q(T)\,T^n\,v = 0$ for all $n \geq 1$. This will certainly satisfy the order-13 recurrence

$$\sum_{i=1}^{13} r_i\,c(i+n) = \sum_{i=1}^{13} r_i u\,Q(T)\,T^{n+i} v = u\,Q(T)\,R(T)\,T^n\,v = u\,P(T)\,T^n\,v = 0$$

where r_i are the coefficients of $R(t)$. To show all $c(n) = 0$ it suffices to show $c(1) = ... = c(12) = 0$.

First we compute $w = u\,Q(T)$, then multiply it with the already-computed $T^n v$.

```
> UT[0]:= u:
  for n from 1 to 9 do UT[n]:= UT[n-1].T od:
  w:= add(coeff(Q(t),t,j)*UT[j],j=0..9):
> seq(w . TV[n],n=1..12);
```

0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0 **(5)**

This completes the proof.